SAS Macro for the generation of minimally changed run sequence for Central Composite Designs

Eldho Varghese[1*], Arpan Bhowmik, Seema Jaggi and Cini Varghese
ICAR-Indian Agricultural Statistics Research Institute, New Delhi
[1]ICAR-Central Marine Fisheries Research Institute, Kochi
Email: eldho.varghese@icar.gov.in

Second Order Response Surface Designs (SORDs) are used to explore relationship between the response variable and the input variables and to find out the optimum input combinations to achieve a desired response. Experiments where changing of input factor levels are very difficult/ costly, instead of using the standard form of run sequences, response surface design with less factor level changes in run sequences should be used. The following SAS macro has been developed to generate minimally changed run sequence for Central Composite Designs. Here, user needs to enter the number of factors as 'k'. If the user executes the program after entering the value of k, then the SAS Macro will produce minimally changed run sequence for central composite design along with the number of factor-wise level changes, total changes and the number of runs.

```
/*Program to generate minimally changed CCD*/
%let k=2; /*Enter the number of input factors*/
ods rtf  file= 'output.rtf' startpage=no;
proc iml;
if &k<2 then do;
print 'Enter a number >=2';
end;
else do;
n0=round((4*(sqrt(2**&k)))+4-(2*&k));
n=(2**&k)+(2*&k)+n0;
alpha=((2**&k)**(1/4));
*print n0 n alpha;
centre=j(n0,&k,0);
ax=alpha//(-alpha);
axial= i(&k)@ax;
axial1=j(nrow(axial)-2,ncol(axial),0);
do i=1 to nrow(axial1);
axial1[i, ]=axial[i, ];
end;
*print axial1;
axial_last=j(2,&k-1,0);
axial_last=axial_last||ax;
*print axial_last;
*print axial centre axial_last;
s=j(&k,1,2);
a=j(max(s),nrow(s),0);
do kk=1 to nrow(s);
```

```
m=mod(s[kk, ],2);
do i=1 to s[kk, ];
do j=i to s[kk, ];
if m=1 then
          do;
          a[j,kk]=-((s[kk, ]-1)/2)+(i-1);
          end;
          else
          do;
          if -(s[kk, ]/2)+(i-1)<0 then do;
          a[j,kk]=-(s[kk, ]/2)+(i-1);
          end;
          else do;
          a[j,kk]=-(s[kk, ]/2)+i;
          end;
          end;
end;
end;
end;
*print a;
aa=j(s[1, ],1,0);
do i=1 to s[1, ];
aa[i,]=a[i,1];
end;
*print aa;
sum=1;
do j=1 to nrow(s)-1;
do i=1 to nrow(aa);
kk=repeat(aa[i,],s[j+1,],1);
if i=1 then do;
aaa=kk;
end;else do;
aaa=aaa//kk;
end;
end;
*print aaa;
sum=sum*s[j, ];
if mod(sum,2)=0 then do;
ggg=j(s[j+1, ],1,0);
do i=1 to s[j+1, ];
ggg[i,]=a[i,j+1];
end;
ggg1=ggg*-1;
ggg2=ggg//ggg1;
hh=repeat(ggg2,sum/2,1);
aa=aaa||hh;
end;
```

```
else do;
ggg=j(s[j+1, ],1,0);
do i=1 to s[j+1, ];
ggg[i,]=a[i,j+1];
end;
ggg1=ggg*-1;
ggg2=ggg//ggg1;
hh1=repeat(ggg2,(sum-1)/2,1);
hh=hh1//ggg;
aa=aaa||hh;
end;
end;
*print aa;
Minimal_CCD=aa//axial1//centre//axial_last;
runs=nrow(Minimal_CCD);
print 'Minimally changed run sequences for Central Composite
Design';
print Minimal_CCD;
Factor_Change=j(1,&k,0);
do i=1 to &k;
Factor_Change[1,i]=(2**(i-1));
end;
do i=1 to &k;
if i=1 | i=&k then do;
Factor_Change [1,i]=Factor_Change [1,i]+3;
end; else do;
Factor_Change [1,i]=Factor_Change [1,i]+4;
end;
end;
Total_Change=sum(Factor_Change);
print 'Factor-wise number of changes in the run sequence';
print Factor_Change;
print 'Total number of changes in the    run sequence';
print Total_Change;
print 'Number of Runs';
print runs;
end;
ods rtf close;
quit;
```

SAS Output

Minimally changed run sequences for Central Composite Design

Minimal_CCD	
-1	-1
-1	1
1	1
1	-1
1.4142136	0
-1.414214	0
0	0
0	0
0	0
0	0
0	0
0	0
0	0
0	0
0	1.4142136
0	-1.414214

Factor-wise number of changes in the run sequence

Factor_Change	
4	5

Total number of changes in the run sequence

Total_Change
9

Number of Runs

runs
16

Research Paper:

- Eldho Varghese, Arpan Bhowmik, Seema Jaggi, Cini Varghese and Shwetank Lall, (2019). On the Construction of Response Surface Designs with Minimum Level Changes. *Utilitas Mathematica*, 110, 293-303.
- Eldho Varghese*, Arpan Bhowmik, Seema Jaggi, Cini Varghese, Charanjit Kaur (2017). On the generation of cost effective response surface designs, *Computers and Electronics in Agriculture*, 133, 37-45.

- Eldho Varghese*, Arpan Bhowmik, Seema Jaggi, Cini Varghese, Charanjit Kaur (2020). Corrigendum to “On the generation of cost effective response surface designs” [Comput. Electron. Agric. 133 (2017) 37–45], *Computers and Electronics in Agriculture*, 170, 105272.